

Open Geospatial Consortium Inc.

Date: 2011-03-01

Internal reference number of this document: OGC 09-047r3

External identifier of this document: <http://www.opengis.net/doc/POL-NTS/DOC/1.2>

Version: 1.2

Category: OpenGIS® Policy

Editor: Simon Cox

Name type specification - documents

Copyright © 2011 Open Geospatial Consortium
To obtain rights of use, visit <http://www.opengeospatial.org/legal/>.

Warning

This document defines an OGC Policy. It is subject to change without notice. This document is an official position of the OGC membership on this particular topic.

Contents	Page
1 Scope.....	1
2 Normative references:.....	2
3 Naming rule	3
3.1 OGC name schemes.....	3
3.2 Production rule for document names	3
4 Name assignment policy.....	5
4.1 Document types	5
4.2 Document element types.....	5
4.3 Names	5
5 Examples.....	6
5.1 Complete documents.....	6
5.2 Document elements.....	6

Document type: OpenGIS® Policy
Document subtype: Name Type Spec
Document stage: 1.1
Document language: English

i. Preface

This document specifies a rule for constructing OGC names that may be used for identifying documents and elements within a document.

ii. Document terms and definitions

This document uses the normative terms (SHALL, SHOULD, etc) defined in Subclause 5.3 of [OGC 06-121r3], which is based on the ISO/IEC Directives, Part 2: Rules for the structure and drafting of International Standards. In particular, the word “shall” (not “must”) is the verb form used to indicate a requirement to be strictly followed to comply with this specification.

Name production rules in this document are expressed using ABNF (IETF RFC 5324).

iii. Revision History

Date	Internal version	Editor	Sections modified	Description
25 March 2009	0.1.0 Draft	Simon Cox	N/A	Initial Draft Document.
1 April 2009		Simon Cox	All	Remove specification elements from scope of 'doc' branch. Minor tweaks and corrections for consistency with the other OGC-NA documents.
21 May 2009		Simon Cox	3	Replaced EBNF with ABNF
January 2010	1.1	Simon Cox	All	ABNF revised to match RFC 3986; http URI syntax made explicit
November 2010	1.2	Simon Cox	3.2, 5.2	'#' allowed as well as '/' before doc element in http URI form. '-' allowed as well as '/' to separate sub-elements of doc element.

Name type specification - documents

1 Scope

An OGC name may be provided for a *document* published by OGC, such as OpenGIS Standards, Best Practice Papers, Discussion papers, and also for an individual element within document, such as clauses, tables, or figures.

The precise scope of documents that may be identified with OGC Names is provided by the set of items in the register <http://urn.opengis.net/register/ogc-na/doc-type>. The precise scope of document elements that may be identified with OGC Names is provided by the set of items in the register <http://urn.opengis.net/register/ogc-na/doc-element-type>.

2 Normative references:

IETF RFC 2141 URN Syntax (1997) <http://tools.ietf.org/html/rfc2141>

IETF RFC 2616 Hypertext Transfer Protocol -- HTTP/1.1 (1999)
<http://tools.ietf.org/html/rfc2616>

IETF RFC 3986 Uniform Resource Identifier (URI): Generic Syntax (2005)
<http://tools.ietf.org/html/rfc3986>

IETF RFC 4395 Guidelines and Registration Procedures for New URI Schemes (2006)
<http://tools.ietf.org/html/rfc4395>

IETF RFC 5141 A Uniform Resource Name (URN) Namespace for the International Organization for Standardization (ISO) (2008) <http://tools.ietf.org/html/rfc5141>

IETF RFC 5165 A Uniform Resource Name (URN) Namespace for the Open Geospatial Consortium (OGC) (2008) <http://tools.ietf.org/html/rfc5165>

IETF RFC 5234 Augmented BNF for Syntax Specifications: ABNF (2008)
<http://tools.ietf.org/html/rfc5234>

OGC 05-020r10, Technical Committee Policies and Procedures
<http://www.opengis.net/doc/POL/PnP>

OGC 09-046r2, OGC Naming Authority – Procedures
<http://www.opengis.net/doc/POL/OGC-NA>

3 Naming rule

3.1 OGC name schemes

Two URI schemes [IETF RFC 3986] are defined by OGC to provide persistent names for resources of interest in geographic information infrastructures. The generic syntax for OGC names is described in [OGC Naming Authority – Procedures].

The generic syntax for OGC http URIs is

```
URI = "http://www.opengis.net/" OGCResource "/" ResourceSpecificPath
```

The generic syntax for OGC URNs is [IETF RFC 5165]

```
URN = "urn:ogc:" OGCResource ":" ResourceSpecificString
```

The following ABNF adapted from [IETF RFC 3986] provides some basic definitions required in the rest of this document.

```
segment      = *pchar
segment-nc   = *pchar-nc
segment-nz   = 1*pchar
segment-nz-nc = 1*pchar-nc

pchar        = unreserved / pct-encoded / sub-delims / ":" / "@"
pchar-nc     = unreserved / pct-encoded / sub-delims / "@"
pct-encoded  = "%" HEXDIG HEXDIG
unreserved  = ALPHA / DIGIT / "-" / "." / "_" / "~"
reserved    = gen-delims / sub-delims
gen-delims  = ":" / "/" / "?" / "#" / "[" / "]" / "@"
sub-delims  = "!" / "$" / "&" / "'" / "(" / ")"
             / "*" / "+" / "," / ";" / "="
```

3.2 Production rule for document names

An OGC name for a document or document element shall be produced using the following rule:

```
OGCResource   = "doc"

ResourceSpecificPath = doc-type "/" doc-identifier [ ( "#" / "/" )
doc-element]
```

ResourceSpecificString = doc-type ":" doc-identifierURN [":" doc-elementURN]

doc-type = segment-nz-nc; token from the register of OGC document types¹

doc-identifier = doc-name ["/" doc-version]

doc-identifierURN = doc-name [":" doc-version]

doc-name = segment-nz-nc ; acronym or code for document topic

doc-version = 1*DIGIT "." 1*DIGIT ["." 1*DIGIT]

doc-element = doc-element-type ("-" / "/") element-number

doc-elementURN = doc-element-type ":" element-number

doc-element-type = segment-nz-nc; token from the register of OGC document element types²

element-number = 1*DIGIT *("." 1*DIGIT)

NOTE: The document element (clause, table, figure, etc) is optional, If present, in the http URI form the element designator follows either a '/' or '#', and sub-elements may be separated by either '/' or '-'. If the '#' form is used as a fragment separator resolution of the secondary resource is delegated to the http client. If the '#' form is used as a fragment separator, the '-' is recommended as the separator for sub-elements.

¹ <http://uwww.opengis.net/register/ogc-na/doc-type>.

² <http://uwww.opengis.net/register/ogc-na/doc-element-type>

4 Name assignment policy

4.1 Document types

The registers of document types <http://www.opengis.net/register/ogc-na/doc-type> is controlled by OGC-NA. Changes to this register (additions, deletions, and supersession) shall be initiated by a submission to the OGC Naming Authority names@opengeospatial.org.

4.2 Document element types

The registers of document element types is controlled by OGC-NA. Changes to this register (additions, deletions, and supersession) shall be by application to ogcna@lists.opengeospatial.org.

The current state of the register of document element types is shown at <http://www.opengis.net/register/ogc-na/doc-element-type>

4.3 Names

The register of names <http://www.opengis.net/register/ogc-na/name> is controlled by OGC-NA. Changes to this register (additions, deletions, and supersession) shall be initiated by a submission to the OGC Naming Authority names@opengeospatial.org.

NOTE: The approval of any new public document by OGC shall automatically trigger a registration request to OGC-NA. The name assigned shall be included on the cover page of the document.

5 Examples

5.1 Complete documents

Example 1 The OGC Name for the WMS specification, version 1.3, is:

http URI Form

```
http://www.opengis.net/doc/IS/WMS/1.3
```

URN Form

```
urn:ogc:doc:IS:WMS:1.3
```

Example 2 The OGC Name for the Discrete Coverage Interleaved Encoding best practice document, version 0.2.0 is:

http URI Form

```
http://www.opengis.net/doc/BP/DCV/0.2
```

URN Form

```
urn:ogc:doc:BP:DCV:0.2
```

Example 3 The OGC Name for the KML Requirements engineering report, version 1.0 is:

http URI Form

```
http://www.opengis.net/doc/ER/KML-REQ/1.0
```

URN Form

```
urn:ogc:doc:ER:KML-REQ:1.0
```

Example 4 The OGC Name for the charter for the Hydrology Domain Working Group is:

http URI Form

```
http://www.opengis.net/doc/CHARTER/HYDRO-DWG
```

URN Form

```
urn:ogc:doc:CHARTER:HYDRO-DWG
```

5.2 Document elements

Example 5 The OGC Name for Clause 5.2 of the WMS specification, version 1.3, is::

http URI Form

```
http://www.opengis.net/doc/IS/WMS/1.3#clause-5.2
```

URN Form

```
urn:ogc:doc:IS:WMS:1.3:clause:5.2
```

Example 6 The OGC Name for Figure 1 of the WMS specification, version 1.3, is::

http URI Form

`http://www.opengis.net/doc/IS/WMS/1.3#figure-1`

URN Form

`urn:ogc:doc:IS:WMS:1.3:figure:1`

Example 7 The OGC Name for Annex A.2.5 in the Modular Specifications standard, version 1.0, is::

http URI Form

`http://www.opengis.net/doc/std/spec/1.3#clause-a.2.5`

URN Form

`urn:ogc:doc:std:spec:1.0:clause:A.2.5`